# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-x11-protocol
pkgver=0.56
pkgrel=13
pkgdesc="Perl/CPAN Module X11::Protocol : Raw interface to X Window System servers"
arch=("any")
url="https://search.cpan.org/dist/X11-Protocol"
license=("GPL" "PerlArtistic")
depends=("perl")
source=("https://www.cpan.org/authors/id/S/SM/SMCCAM/X11-Protocol-$pkgver.tar.gz")
sha256sums=('de96dd6c7c1f25f3287aa7af64902bf84acaaa8e0c3bb76aa1676367e04a08b7')

build() {
  cd "$srcdir"/X11-Protocol-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/X11-Protocol-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
